"YOUR LIFE WILL NEVER BE THE SAME COMPUTER CLUB"

Vol.1. No.5

Finally the fifth newsletter is out. From the letters we have received, it is quite obvious that most of our clubmembers are hooked on the newsletter (luckily it is legal!). Why only a month after the fourth newsletter was circulated, we were receiving letters demanding to know when the fifth one would be done.

Well here it is, and as you can see it has changed a bit again. There are now 4 pages of consumer programs, a new section called "For Your Information", and the second part of our series on graphics. Have fun reading this issue clubmembers, we will be waiting for your responses.

Editor

SPOTLIGHT ON SOFTWARE

Electronic Files - MA425

Are you tired of searching through drawer after drawer, file after file, many times not finding the information you were looking for?

End those problems with APF's Electronic Files program. With APF's Electronic Files program, finding information is as easy as typing in a file name and seconds later, having the information you have requested appear on the screen in front of you. Save time and aggravation; make your **life easier** with APF's Electronic Files program.

Bar Charts - MA350

Who has the best way of displaying sales projections, cost analysis, or comparision charts? That's right! APF does and so can you with the APF Color Bar Chart program.

APF's Bar Chart program is designed to display your data in color ful, easy to read chart form. It includes the options of displaying one chart, two charts side by side, or two charts combined. With APF's Colorful Bar Chart program, conducting financial business can be fun.

SPOTLIGHT ON HARDWARE

ATTENTION SMALL BUSINESS PEOPLE!

APF is proud to introduce a new member of its family - The Imagination Machine II (from here on known as IM-2). Now you can perform your business operations in a fraction of the time it once took, have complete organization of accounts, and save money in overhead; all while discovering the world of computers!

Don't worry, a degree in Computer Science is not required. A little common sense and some practice using the machine is all that is necessary.

For those of you technically minded, read over the specifications and take notice of all of the features and hardware that are included for an affordable price of $1599.95!!

27K RAM (Random Access Memory)
- 11K Direct Access in Basic Unit
- 16K Additional Direct Access (plug-in cartridge form)

ROM (Read Only Memory)
- 14K Bytes

Keyboard
- 53-key Stepped Typewriter Style

Connections

Any Standard Color or Monotone or Video Monitor (optional APF Model TVM-10)

3 Major Screen Display Modes

32 Characters X 16 Lines Alpha Numeric Mode, which can be intermixed with 64 X 32 (in 8 colors) semigraphics mode.

Object oriented high resolution graphics - 128 user definable objects, giving resolutions of 256 X 192 (in 2 colors) or 128 X 192 (in 8 colors).

Point by point high resolution graphics - 128 X 96 points definable in 4 colors.

Tape Deck

Quality built-in cassette tape deck which uses standard Philips type cassettes, transfers computer programs at 1500 BAUD rate. Audio record/playback capability. Speaker, volume control, microphone jack built-in.

User Programmable

Programmable in "Basic" language.

Programmable in MC6800 language.

Additional Features

R.F. modulator included for connection to T.V.

Disk operation system (DOS) contained in ROM.

Built-in character generator ROM for use with object graphics mode - allows upper/lower case capability.

Building Block - includes FI-100 (mini-floppy disk interface)

Parallel Printer Interface

Dual Floppy Disk Drives

All necessary cables, power supplies, technical and operational manuals.

This system has been designed with all the necessary hardware and it will function with programs in floppy disk or cassette form. (Please note that almost all of the IM-1's hardware and software is compatible with the IM-2, with the exception of the Backgammon and the Baseball programs).

You get all this hardware for only $1599.95! It's unbelievable, but it's true. Leave it to APF to make your dreams of owning a computer which satisfies your needs, a little more of a reality.

Level 2 Basic

Now perhaps the excitement of the IM-2 is under control (doubtful), but there is still another reason for your adrenaline to be pumping. The Level 2 Basic is now here!!! (please note: There are two versions of Level 2 Basic; one for the IM-1 and the other for the IM-2. Please specify your machine when ordering Level 2 Basic.) Due to a limited amount of space, we will only do a quick summary of the Level 2 Basic. Further information can be obtained from APF directly.

Features of the Level 2 Basic

Keyboard (Refers to IM-2 only)

Operates in both upper and lower case. Control Key (CTRL) and various other keys enter graphic symbols, Pi (π), superscripts, and more.

System Commands

List - lists your program one line at a time; from one specified line number to another, from one specified line number to the end of the program, or lists complete program.

Delete - operates in the same manner as the list command, except it deletes line statements instead of listing them.

Clear - clears values of variables.

CLS - clears screen and "HOMES" the cursor.

New - clears out all line statements and variables from computer's memory.

Editor

Used for editing stored and currently entered line statements by inserting and deleting characters. Cursor can be moved throughout the line and sent instantly to the beginning of the line.

Math

9 digit floating decimal point with scientific notation.
Standard operators (+,-,/,*,↑)

Variables

Can be numerical or string type and does not require dimensioning.

Tape

CSAVE and CLOAD programs to tape.

Statements

PRINT, STOP, END, CALL, PEEK, POKE, REM, LET, INPUT, DATA, NEW, RESTORE, FOR, NEXT, STEP, TO, GOSUB, RETURN, GOTO, ON, IF ERROR, PRINT@, IF_THEN_ELSE, PRINT USING, TIME, DEFINE FUNCTION, CONTINUE

Functions

SIN, COS, TAN - Arguement in Radians, ARC TAN, LN, EXP, X↑Y, SQR, TAB, INT, ABS, RND, SGN, STR$, VAL, RIGHT$, LEFT$, MID$, A$+B$, IN KEY$, LEN, ASC, CHR$

Graphics

GR0 Alpha/Semigraphics mode
GR2 Point x Point 128 X 96 resolution in 4 colors (IM-2 only)
GR4 Object Mode - 128 objects (IM-2 only)
GR6 Point x Point 256 X 96 resolution in 2 colors (IM-2 only)

Sound

Sound X,Y Frequency X, Duration Y

This Level 2 Basic is retailing for $99.95, and comes in cartridge form. A great addition to your Imagination Machines, at an unbeatable price!!!

FOR YOUR INFORMATION

Stand back and make way for our new section, "For Your Information". In this section we will pass on bits of information which we have received from our readers and staff members. We hope that this section proves to be both informative and helpful.

The first bit of information concerns the Newsletter. It has been brought to our attention that many YLWNBS clubmembers expect newsletters monthly. Let it be known that this is not to be expected. To lessen the anxiety of those believing they've missed issues, and to inform still more of you that are uncertain, the newsletter is produced quarterly.

Now that that little item is cleared up we can move on to substitutes for logical "OR" and "AND" functions in IF-THEN statements.

Although the logical "OR" and the logical "AND" functions are not in the "Basic" vocabulary, there is a way to compensate for them. For instance, the "AND" function would be used as follows:

IF A=5 AND B$= "FRED", THEN....
On the Imagination Machine it is written as:

IF A=5 THEN IF B$= "FRED" THEN..
Now although a replacement for the "OR" function is not as obvious, there is one and it uses the multiplication property of zero. For example, the "OR" function would be used as follows:

IF A=6 OR B=5 OR C=4 THEN 1000
On the Imagination Machine it is written as:

IF (A-6)*(B-5)*(C-4)=0 THEN 1000
Our thanks to Bruce Chapman for this information.

Now for all you hobbyists: APF is offering a video conversion kit (Model VK1) which will enable you to connect your Imagination Machine to any black and white monitor. To convert your machine requires new connections in the MP1000 section, which can be done by you or APF.

The kit sells for $15.00 complete with schematics, and may be purchased from APF directly. If you choose to send in your unit to APF for converting, please add $5.00 for postage and handling.

Another new APF item is a Diagnostics Program. APF now offers a set of diagnostics which will check Graphics, ROM, RAM, KEYBOARD, and TAPE functions. This program is available on cassette tape for $15.00 and may be obtained from APF directly. (Model #SY-50).

Our final bit of information should interest many clubmembers. In approximately two months, APF will have a complete user written software book available. This book will provide users with an excellent exchange of ideas and programs, ranging from business applications to renumbering of programs. For more information contact the Editor at APF.

That's it for this issue's "For Your Information". If you have any thoughts or ideas which you would like to share with fellow clubmembers, don't hesitate to send them in.

YOU ASKED FOR IT

This issue's "You Asked For It" will undoubtably bring cheers from many of our clubmembers for the following three reasons: Sound Generation, Chaining Programs, and information on Machine Language. Let the noise begin with Sound Generation!!

A number of requests have been made asking for Call numbers for Sound Generation. Here are five such Call numbers, along with a program which demonstrates some of the remarkable sounds which can be made with your Imagination Machines.

CALL NUMBERS	PROGRAM
17001	5 DIM A(5)
17005	20 For I=1 TO 5:READ
17007	A(I): NEXT I
17017	30 DATA 17001,17005,
17035	17007,17017,17035
	40 FOR I=1 TO 1000:
	J=INT (6*RND(0))
	55 IF J=0 THEN J=J+2
	60 K=INT (4*RND(0))
	70 FOR L=0 TO K
	80 CALL A(J)
	90 NEXT L
	100 NEXT I

And now for chaining programs! For those of you that have disk systems, you are aware that the Disk Operating System allows "chaining" of programs. Many of you have made attempts at chaining programs and were not successful. Rest assured that the following information will produce successful results.

Chain disk programs, or in other words, have a program residing in the computer's memory, load another program from the diskette and automatically have it run. The command structure is simply to have a program statement which says: RUN "FILE NAME". When that statement is executed, "FILE NAME" is loaded from diskette and automatically started.

Now before you run off and try to chain programs, there is a slight problem which must be corrected. Pointers which tell "BASIC" where to dimension variables are not initialized correctly when a program is chained in. A very simple solution is as follows. The first line statement of each program being chained must be: POKE 41009,PEEK (41984): POKE 41010, PEEK (41985). With this as your first line statement in programs being chained, you should have no problems at all.

On to the information concerning machine language. It's good to know that many of our "BASIC" programmers

are making an effort to learn machine language programming. To make this process a little easier, we have compiled a list of books which may be helpful. Good Luck!!

1. Motorola 6800 Programming Reference Manual-Motorola Semiconductor Products Box 20912, Phoenix, Arizona 85036

2. "68" Micro-Journal-3018 Hamill Road, PO Box 849, Hixson, Tenn. 37343

3. Basic Microprocessors and the 6800 Motorola Semiconductor Group by Ron Bishop

4. The 6800 Microprocessor:A Self Study Course with Applications by Lance Leventhal

HIGH RESOLUTION GRAPHICS

Judging by your responses it appears that the section on high resolution graphics was quite helpful. In this issue we will discuss high resolution graphics using four colors (in two color sets) in the 128 x 192 mode. This mode is a bit more difficult but then again, it's a lot more colorful and quite useful. To keep your enthusiasm up, just keep reminding yourself that it is the mode used in APF's Baseball, Backgammon, and UFO cartridges.

The first step is to get your fourth newsletter out, because a lot of the information that you learned for the 256 x 192 mode is applicable to the 128 x 192 mode; the screen map, and the general program for putting codes into memory are the same. The two major differences are the POKE commands to enter this mode, and the method of defining the colorful shapes.

First the easy part-the pokes to enter the 128 x 192 mode are: POKE 8193,56 and POKE 8194,158. Easy right?!

Now for defining the shapes- In the 128 x 192 mode, each shape is four dots wide (as opposed to 8 dots in the 256 mode) and sixteen lines high. Each dot can be selected to be one of four colors. Once again this requires the use of Binary. For four possible colors we need two Binary digits (bits) in order to indicate the four colors. These bit pairs are; 00,01,10,11. Each eight bits (commonly known as a byte) has four pairs, which works out to be four colored dots per byte The basic color set and the four bit pairs are:

Bit Pair	Color
00	Green
01	Yellow
10	Blue
11	Red

Getting a little complicated now, so pay close attention. A shape (16 lines of bytes) should be defined graphically. A code for each line should be generated first in Binary, then Hexidecimal and finally converted to Decimal, since we want to program in Basic. As a simple example (kidding aside) lets say we want a shape with all sixteen lines having all four dots yellow. Graphically this appears as:

Bit Pair#	4321	Binary	Hex	Decimal
Any Line	YYYY	01010101	55	85

Still keeping it simple lets define a shape in all green.

Bit Pair#	4321	Binary	Hex	Decimal
Any Line	GGGG	00000000	00	00

Keep in mind that for each shape there are sixteen lines. We have kept it simple so all sixteen lines of each shape have the same color. Now it is time to incorporate all of this information into a program. The following program will put up alternate rows of green and yellow boxes on the screen.

```
  5 REM SET GRAPHICS MODE
 10 POKE 8193,56:POKE 8194,158
 20 REM LINES 30-70 MOVE OBJECT
    CODES TO MEMORY
 30 DATA 85,85,85,85,85,85,85,85,85,
    85,85,85,85,85,85,85:REM SHAPE#0
```

```
 40 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,
       0,0,0:REM SHAPE#1
 50 FOR I=512 to 543
 60 READ A:POKE I,A
 70 NEXT I
 80 REM LINES 90-160 SETUP SCREEN
    MAP WITH ALTERNATE ROWS HAVING
    SHAPE 0 AND SHAPE 1
 90 DATA 0,32,1,32,0,32,1,32,0,32,
       1,32,0,32,1,32,0,32,1,32,
       0,32,1,32,999,999
100 X=0
110 READ A,B:IF A=999 THEN 170
120 FOR J=1 TO B
130 POKE X,A
140 X=X+1
150 NEXT J
160 GOTO 110
170 REM SCREEN IS SET UP, WAIT FOR
    KEY TO BE PRESSED BEFORE RETURN
    ING TO REGULAR MODE
180 IF KEY$(0)="" THEN 180
190 POKE 8194,60:STOP
```

You should not have had any trouble with the above, but if you did recheck your program carefully and reread both graphic writeups. The shapes you define of course can be more complex and you can mix the four colors on a line. In the next issue we will discuss the alternate color sets (what more colors???) and vertical motion of objects.

NEW SOFTWARE

General Ledger - BP-50
Requires: IM-1 or IM-2
BB-2
Dual Disk Drives
R8K Memory Module-R16K for IM-2
80 Column Printer (appropriate cable and interface)

APF's General Ledger program gives you all the necessary tools for maintaining an accurate and complete general ledger. It will produce all the standard financial statements, access all account and posting information, and several additional reporting functions. This program will maintain 2034 posting entries per disk, and will verify all entries to avoid processing postings which are out of balance. Seven accounts may be entered within each entry, reference numbers are automatically assigned, and a description for the entry is an optional data entry. There is also a balance forwarding function which provides standard monthly closing procedures, plus reversal processing and year-end closing entries. This program retails for $199.95, and may be purchased from APF directly.

Compu-Calc - BP-60
Requires:IM-2
BB2
R16K Memory Module
Dual Disk Drives
80 Column Printer (appropriate cable and interface)

Compu-Calc is a program which combines the functions of a calculator, paper and pencil, with a computer's memory and speed. This program is essential for people that deal with numbers. This includes people who plan budgets, compare actual results to budgeted forecasts compute financial ratios and tax consequences, commission rates, sales costs, advertising expenditures, etc. Compu-Calc's complete with features such as: Fix-point arithmatic, printouts for discussion and documentation, and most importantly, on the screen viewing of all calculations and comparisons. This program retails for $129.95 and can be obtained directly from APF.

LETTERS TO THE EDITOR

Once again we get the chance to show you the wonderful letters we receive from our clubmembers. These letters are our proof that the APF Computer Club is a true success!!

I will be attending school next year at St. Paul's in Concord, New Hampshire. Throughout the year I will be able to take short trips to nearby cities such as N.Y. On one of these occasions I would like to arrange to visit you if it's convenient. Larry Drebes

We have received a number of similar requests and we are more than

happy to meet with our clubmembers. If you would like to visit with us, contact the Editor so that a convenient time may be arranged.

I was very pleased to receive the Computer Club Newsletter, especially as I live as far away as Australia. Your letter and the newsletters that I have received have certainly served to reduce the sense of "computer isolation" that I feel here. The backup shown by the company through this club is tremendous!

David Powell

If a registration card is sent in you immediately become a member of our club, no matter where you live!

I have owned the IM-1 for approx. one year and a few people I know have had one for awhile. We have gotten together a club, Compu-Swap, for users of the APF Imagination Machine. We would like our club to grow and hopefully you can help us. I would appreciate it if you could put the following announcement in your next newsletter. Thank you very much.

Bill Vackner

The following is a copy of the announcement Bill Vackner sent in to APF.

COMPU-SWAP PO BOX 1371, WEST CALDWELL, NJ 07006

COMPU-SWAP is a group of users, who for the past 6 months have operated an APF Imagination Machine Club for the purpose of exchanging programs and general information about our computers. COMPU-SWAP would like to extend an invitation to other APF owners to join our club. Anyone interested should send a SASE to the above address.

That's it for now clubmembers. Once again, thanks for your involvement and support.

BUGS

Oh those nasty bugs are back again. But never fear, there is a solution for all of them. The first set of bugs are merely typographical errors our clubmembers have discovered. For instance:

Basic Tutor corrections

Page 160

4th para. delete five rows,
insert six rows

5th para. delete (5x8)
insert (6x9)
delete 40
insert 54

7th para. after DIM B$ (4,12)
delete four words
insert five words

insert sentence: with letter arrays, the second number exactly equals the memory space required.

Page 29- The very last item, A3$ is not incorrect as suggested. It should read 3A$.

Page 36- Line 70-The monitoring program will not accept the space between 0=NO, and 1=YES. Therefore delete the space.

Page 116- Line 50-There should be 8 spaces between interest and total, not one as shown.

Page 228 under D-DATA-delete the quotation marks around data items.

Thank you David Powell for the Basic Tutor corrections.

Fourth YLWNBS Newsletter Corrections

In section B in the High Resolution Graphics write-up, remove one zero from line 60.

In the D section, insert the following: 130 GOTO 90.

The final bug in the 5th newsletter is in the bug section. Stop laughing and insert in number six; press the CL key and then the Break key.

Now for the serious bugs:

RND-We have discovered a possible condition which occurs in powering up, which always makes the RND (0) function produce a result of zero. Since each time you power up this

condition may or may not occur, the best solution is to place the following statement in any program which uses the RND function: POKE 40994,27. This statement will alleviate your problems.

Disk-There is a problem which occurs when you're trying to delete disk files that were named by a string variable value. Ex. DIM A$ (7)="TEST":OPEN 0 A$.
Now if you try to delete "TEST" by typing SAVE"TEST"K, the error message, "A FILE NOT FOUND" will occur. The reason for this is that on the diskette directory the file named "TEST" is actually seven characters. (TEST being the first four and the last three are ASCII code 0). The solution for this problem is to type: SAVE"TEST CTRLA CTRLA CTRLA CTRLA"K. (For CTRL press the control key and the "A" key simultaneously, leaving out all of the spaces in between the words.) By typing in the above you add three zeros to the name "TEST", thus eliminating your problem.

The final bug was sent in by David Powell one of our Australian clubmembers! David has a knack for discovering bugs and we are certainly thankful to him for the following:
Spelling Duel:If at the end of a game the operator selects new game/ new players, then the old player(s) name(s) and score(s) are not cleared
The fix for this:

```
   0POKE 40994,27
   1POKE 24578,38
2010DIM SC (3,6),NM$ (4,10)
2140FOR P=1 TO NP:NM$(P-1,0)="TEN
    SPACES":FOR Z=0 TO 6:SC(P-1,0)
    =0:NEXT
2160INPUT NM$(P-1,0):IF ASC(NM$(P-
    1,0))=32 MUSIC"/1":GOTO 2150
2170MUSIC "7":NEXT:NG=1
```

That's all of the bugs we are aware of. If you have discovered any other bugs, be a loyal clubmember and let the rest of us know about it.

SUBMITTED CLUBMEMBER PROGRAMS

This first program is a simulation of a moon landing sent in to us by Bruce Neustater of Morganville N.J. The object of game is to land the spaceship on the landing pad without crashing. Try it, but believe me it is not easy!

```
1  CALL 17046: DIM A$(1): POKE 24578,38: GOTO 6
2  PRINT "ENTER RANK:": PRINT "1=PRIVATE": PRINT "2=MAJOR": PRINT "3=GENERAL": M
USIC "1": PRINT
3 A$= KEY$ (2): IF A$="" THEN 3
4  MUSIC "-4":R=A$:R=R/100000000
5  RETURN
6  CALL 17026
7  GOTO 9
8 A=L*32+P+512: POKE 40960,A/256: POKE 40961,A- INT (A/256)*256: RETURN
9  CALL 17046
10  COLOR =0: SHAPE =15: FOR U3=0 TO 15: HLIN 0,31,U3: NEXT U3
20  DIM A$(1)
30  GOSUB 8
31  POKE 8193,60
35  PRINT "     <<<< MOON LANDER >>>>": PRINT : PRINT "^=THRUST       >=RIGHT TH
RUST              <=LEFT THRUST": PRINT
38  PRINT : PRINT
40  COLOR =4: HLIN 15,20,12
45  COLOR =3: PLOT 17,5: PLOT 16,6: PLOT 18,6: FOR Z=0 TO 250: NEXT Z: COLOR =0:
 PLOT 17,5: PLOT 16,6: PLOT 18,6
46  COLOR =3: PLOT 17,7: PLOT 16,8: PLOT 18,8: FOR Z=0 TO 250: NEXT Z: COLOR =0:
 PLOT 17,7: PLOT 16,8: PLOT 18,8
47  COLOR =3: PLOT 17,10: PLOT 16,11: PLOT 18,11
50  PRINT "HIT ANY KEY TO START"
60  FOR Z=1 TO 150: IF  KEY$ (2)<>"" THEN 100
70  NEXT Z
75  MUSIC "3"
80  GOTO 60
100  CALL 17046: GOSUB 2
101  FOR U3=1 TO 32: PRINT : NEXT : CALL 17026
110  SHAPE =15
120  COLOR =0: HLIN 0,31,15
130  COLOR =4: SHAPE =4: PLOT 2,4: PLOT 7,11: PLOT 16,6
131  PLOT 28,9
135  SHAPE =15
140 O= INT (50/R):X1=17:Y=3:X=4:HS=.2*R:VS=.2*R
145 Y1=14
150  COLOR =4: HLIN X1,X1+3,Y1
200  GOSUB 8
205  PRINT "V.SPEED= ";: IF VS<0 THEN  PRINT "^";
206  IF VS>=0 THEN  PRINT "+";
207  PRINT " "; ABS ( INT (VS*100));"  H.SPEED= ";: IF HS>0 THEN  PRINT "-> ";
211  IF HS<0 THEN  PRINT "<- ";
212  PRINT  INT ( ABS (HS*100))
215  PRINT "      FUEL= ";O: IF O<=20 THEN  POKE 544,76: POKE 545,79: POKE 546,
87: POKE 547,96: POKE 548,97
216  IF O<20 THEN  MUSIC "/-10"
217  PRINT
218  IF E=1 THEN 3000
270 A$= KEY$ (2): IF A$="S" THEN VS=VS-.71: CALL 17006
271  IF Y>10 THEN  IF A$="S" THEN VS=VS+.22
275  IF A$="W" THEN HS=HS+.45
280  IF A$="E" THEN HS=HS+.5
290 R1= RND (0): IF R1>.6 THEN R1=R1/2
295  IF VS*100<50 THEN R1=R1/2
300  IF A$="" THEN VS=VS+R1
305  IF A$="" THEN HS=HS+.15
307  IF A$<>"S" THEN  IF A$<>"" THEN VS=VS+R1
310 O=O-1: IF O<0 THEN 8000
314 Q1=X:C1=Y
315 X= INT (X):Y= INT (Y)
320 K=Y*32+X+512: POKE 40960,K/256: POKE 40961,K- INT (K/256)*256
321  PRINT " ": PRINT "                                    "
322 X=Q1:Y=C1
330  COLOR =4: SHAPE =4: PLOT 2,4: PLOT 7,11: PLOT 16,6
331  PLOT 28,9
341  IF Y>12 THEN 343
342  GOTO 370
343  COLOR =4: SHAPE =15
344  HLIN 17,20,14
370 X=X+HS:Y=Y+VS
371  IF X>29 THEN X=29
375  IF X<2 THEN X=2
380  IF Y<2 THEN Y=2
381  IF Y>12 THEN  IF X+1<X1+4 THEN  IF X-1>=X1 THEN Y=12
385  IF Y>14 THEN Y=14
390  COLOR =3: SHAPE =15: PLOT X,Y: SHAPE =6: PLOT X-1,Y+1: SHAPE =9: PLOT X+1,Y
+1
500  IF Y>=12 THEN  IF X+1<X1+4 THEN  IF X-1>=X1 THEN 2500
600  IF Y>13 THEN  IF X>X1+4 THEN 1000
605  IF Y>13 THEN  IF X<X1-1 THEN 1000
606  IF Y>13 THEN 1000
700  GOTO 200
1000  COLOR =3: PLOT X,Y: PLOT X+1,Y: PLOT X-1,Y: PLOT X,Y+1: MUSIC "/1/1/144"
1010  MUSIC "/-1000000000000"
2000  MUSIC "1112/1/1/1/1/00000": PRINT "THE LANDER HAS CRASHED!!": PRINT "YOU H
AVE DESTROYED A"
2010  PRINT "$626,718,915,621.00 LANDER!!": GOTO 9900
2500 E=1: GOTO 200
3000  MUSIC "11111666*54": PRINT "YOU HAVE LANDED!!!"
3005 E=0
3010  IF VS*100>120 THEN  PRINT "BUT!!!...YOU MADE A "; RND (0)*1000;" FT. CRATE
R!!!": PRINT "EVERYONE IS DEAD!!": GOTO 9000
3020  IF VS*100>90 THEN  PRINT "... MASSIVE DESTRUCTION HAS": PRINT "OCURRED!..
YOU ARE STRANDED ON THE MOON!"
3030  IF VS*100>90 THEN  GOTO 9000
3040  IF VS*100<20 THEN 4000
3050  PRINT "... ROUGH LANDING.. BUT EVERYONE IS O.K.!"
3060  MUSIC "3": PRINT "DAMAGE TO THE MAIN ROCKET IS      HEAVY!!"
3070  PRINT "..ESTIMATED REPAIR TIME IS          "; INT ( RND (0)*100);" DAYS"
3080  GOTO 9000
4000  PRINT "WOW!!! WHAT A LANDING!!": PRINT "EVERYONE IS FINE...."
4010  PRINT "DAMAGE TO THE SHIP IS MINIMAL!"
4020  PRINT "GREAT JOB CAPTAIN!!": GOTO 9000
7999  STOP
```

```
8000  PRINT "YOU ARE OUT OF FUEL!": FOR F4=1 TO 5: CALL 17006: NEXT
8010  PRINT "EVERYONE IS DEAD!!": PRINT : MUSIC "/-10000000": GOTO 9000
9000  PRINT : MUSIC "3"
9090 RA=( ABS (100*HS))+( ABS (200*VS))-0
9101  IF 0<=0 THEN 9900
9105 RA= INT (RA)
9107 RA=1000-RA
9110  PRINT "YOUR RATING IS:   ";RA: MUSIC "33"
9900  PRINT "HIT 'FIRE' TO PLAY AGAIN"
9910  FOR Z=0 TO 500:A$= KEY$ (2): IF A$="" THEN  NEXT Z
9920  IF A$<>"" THEN 9940
9930  MUSIC "3": GOTO 9910
9940  IF A$="!" THEN 100
9950  MUSIC "3": END
9960  POKE 40960,0: POKE 40961,0
9962  FOR I=1 TO 16: PRINT  SPC (32):: NEXT
9964  POKE 40960,1: POKE 40960,0
9966  PRINT  SPC (8):"..MO ON  LANDER..": PRINT : PRINT "USE THE LEFT JOYSTICK":
 PRINT "PUSH IT TOWARDS YOU TO THRUST"
9967  PRINT "PUSH IT RIGHT OR LEFT TO STEER": PRINT "THE LANDER...." PRINT "
  ... GOOD LUCK ... "
9968  PRINT : PRINT "TYPE 'RUN' WHEN OK APPEARS"
9970  POKE 40960,2: POKE 40961,0
```

The next program was sent in by our very active clubmember, Bruce Chapman. The object of the game is to score the highest amount of points by hitting the enemy starship and the clumsy red alien. While trying to get these ships be alert so that the alien doesn't trample you. You will need to play this game awhile before you really get the hang of it. Good Luck!

```
0  DIM A$(1):M=207:MC=1
10  POKE 8193,60: POKE 8194,158: FOR X=512 TO 768: READ Z: POKE X,Z: NEXT X: FOR
 X=0 TO 383: POKE X,0: NEXT X
20 E=382:R=1: POKE E,1: POKE E-1,2: POKE R,3: POKE R+1,4
29 MOVES=0
30 MO=MO+1: POKE M,0: POKE M+1,0: POKE M-1,0:M=M+MC: POKE M,6: POKE M+1,7: POKE
M-1,8: IF (M-R)*(M-E)=0 THEN 300
31 A$= KEY$ (2): IF A$="" THEN 105
35  POKE E,0: POKE E-1,0
40  IF A$="W" THEN E=E-1: IF E<353 THEN E=353
50  IF A$="E" THEN E=E+1: IF E>383 THEN E=383
55  POKE E,1: POKE E-1,2
59  IF MO>15 THEN 104
60  IF A$<>"!" THEN 30
61  CALL 17016
65  POKE E-32,5:P=E-32:PR=-32
70  CALL 17006: GOSUB 250: POKE P,0:P=P+PR: CALL 17006: IF P>31 THEN  POKE P,5:
IF  KEY$ (2)="" IF (P-M)*(P-M-1)*(P-M+1)<>0 THEN 70
75  IF (P-M)*(P-M-1)*(P-M+1)=0 THEN  CALL 17015: MUSIC "-/100": CALL 17006:SE=SE
+.5:MC=-MC: GOTO 104
80  IF  KEY$ (2)="W" THEN  POKE P,0:PR=PR-1: IF P>31 THEN 70
90  IF  KEY$ (2)="E" THEN  POKE P,0:PR=PR+1: IF P>31 THEN 70
95  IF P>31 THEN 70
100  IF (P-R)*(P-R-1)*(P-R-2)=0 THEN  POKE R,67: POKE R+1,68: MUSIC "-/1010+*1":
SE=SE+1
104 MOVES=0
105 MO=MO+1: POKE M,0: POKE M+1,0: POKE M-1,0:M=M+MC: POKE M,6: POKE M+1,7: POKE
 M-1,8: IF (M-R)*(M-E)=0 THEN 300
120 A$= KEY$ (1): IF A$="" THEN 30
125  POKE R,0: POKE R+1,0
130  IF A$="W" THEN R=R-1: IF R<0 THEN R=0
140  IF A$="E" THEN R=R+1: IF R>30 THEN R=30
145  POKE R,3: POKE R+1,4
149  IF MO>15 THEN 29
150  IF A$<>"!" THEN 105
151  CALL 17016
155  POKE R+33,5:P=R+33:PR=32
160  CALL 17006: GOSUB 250: POKE P,0:P=P+PR: CALL 17006: IF P<352 THEN  POKE P,5
: IF  KEY$ (1)="" IF (P-M)*(P-M-1)*(P-M+1)<>0 THEN 160
165  IF (P-M)*(P-M-1)*(P-M+1)=0 THEN  CALL 17015: MUSIC "-/100": CALL 17006:SR=S
R+.5:MC=-MC: GOTO 29
170  IF  KEY$ (1)="W" THEN  POKE P,0:PR=PR-1: IF P<352 THEN 160
180  IF  KEY$ (1)="E" THEN  POKE P,0:PR=PR+1: IF P<352 THEN 160
185  IF P<352 THEN 160
190  IF (P-E)*(P-E+1)*(P-E-1)=0 THEN  POKE E,65: POKE E-1,66: MUSIC "-/1010+*10"
:SR=SR+1
200  GOTO 30
250  POKE M,0: POKE M+1,0: POKE M-1,0:M=M+MC: POKE M,6: POKE M+1,7: POKE M-1,8:
IF (M-R)*(M-E)<>0 THEN  RETURN
300  POKE 8194,30: FOR X=0 TO 15: PRINT : NEXT X: IF M-E=0 THEN  PRINT "THE U.S.
S. ENTERPRISE HAS BEEN  TRAMPLED BY A CLUMSY ALIEN COM- MUNIST!"
310  IF M-R=0 THEN  PRINT "THE ROMULAN VESSEL HAS BEEN DESTROYED BY THE ALIEN CO-
MMY!"
315  PRINT "THE SCORE FOR THE ROMULAN VESSELIS: ";SR
320  PRINT "THE SCORE FOR THE STARSHIP ENTERPRISE IS: ";SE
330  STOP : REM ARK-THIS  REM ARKABLE "STAR TREKKERS" PROGRAM WAS DESIGNED AND P
ROGRAMMED BY BRUCE CHAPMAN.
999  DATA 170,170,170,170,170,170,170,170,170,170,170,170,170,170,170,170,170,106,90
,86,86,86,86,90,102,102,102,86,102,102,102,102,166,170
1001  DATA 171,165,149,149,149,149,165,153,153,153,149,153,153,153,154,170,170,1
70,106,106,106,72,64,64,66,74,64,96,104,170,170,170
1002  DATA 170,170,169,169,169,33,1,1,129,161,1,9,41,170,170,170
1003  DATA 170,170,170,170,170,170,170,106,170,170,170,170,170,170,170,170,170
1004  DATA 170,170,235,190,255,255,255,125,255,195,190,235,235,170,170,170,234,2
34,170,171,171,238,250,254,254,250,170,170,170,234
1005  DATA 170,170,171,171,170,234,234,187,175,191,191,175,170,170,170,171,170,1
70
```

Dan Wilson of Bloomfield IN. sent us this Maxwells Demons program. It is a bit simple but it is still fun to play. The object of this game is to trap molecules coming from the left side, on the right side of the grid. Have fun folks!

```
0  REM ************************
1  REM *****MAXWELLS DEMONS*****
2  REM ************************
3  REM *****DJW R2 04%27%81*****
4  REM ************************
5  POKE 24578,54: CALL 17046: DIM K$(1): GOSUB 60
6 R= RND (R):R= INT (10*R):K$= KEY$ (2): IF K$="" THEN 47
7  COLOR =2: PLOT 15,7: PLOT 15,8: PLOT 16,7: PLOT 16,8:T=T+1
8  MUSIC "#1": COLOR =1: PLOT 15,7: COLOR =2: PLOT 15,7: IF K$="?" THEN 58
9  COLOR =2
10  ON R GOTO 35,37,39,41,43,45,11
11  COLOR =1: MUSIC "#1"
12  PLOT 16,8: PLOT 16,7: PLOT 15,7: PLOT 15,8
13  GOTO 47
14  REM  DETERMINE POSITION
15  COLOR =1: PLOT 16,8:R= RND (R)
16  COLOR =2: PLOT 16,8:R= INT (R*20)
17  REM  FROM R VALUE
18  IF R=1 THEN  LET A=6:B=5
19  IF R=2 THEN  LET A=8:B=6
20  IF R=3 THEN  LET A=10:B=7
21  IF R=4 THEN  LET A=26:B=5
22  IF R=5 THEN  LET A=18:B=7
23  IF R=6 THEN  LET A=25:B=10
24  IF R=7 THEN  LET A=21:B=8
25  IF R=8 THEN  LET A=24:B=7
26  IF R=9 THEN  LET A=22:B=6
27  IF R=10 THEN  LET A=13:B=8
28  IF R=11 THEN  LET A=7:B=8
29  IF R=12 THEN  LET A=9:B=9
30  IF R=13 THEN  LET A=5:B=7
31  IF R=14 THEN  LET A=5:B=10
32  IF R=15 THEN  LET A=26:B=8
33  IF R=16 THEN  LET A=23:B=9
34  COLOR =1: PLOT 15,8: COLOR =2: PLOT 15,8: RETURN
35  PLOT X1,Y1: GOSUB 15
36  LET X1=A:Y1=B: GOTO 11
37  PLOT X2,Y2: GOSUB 15
38  LET X2=A:Y2=B: GOTO 11
39  PLOT X3,Y3: GOSUB 15
40  LET X3=A:Y3=B: GOTO 11
41  PLOT X4,Y4: GOSUB 15
42  LET X4=A:Y4=B: GOTO 11
43  PLOT X5,Y5: GOSUB 15
44  LET X5=A:Y5=B: GOTO 11
45  PLOT X6,Y6: GOSUB 15
46  LET X6=A:Y6=B: GOTO 11
47  COLOR =4: PLOT X3,Y3: PLOT X5,Y5: PLOT X1,Y1: MUSIC "123"
48  COLOR =3: PLOT X2,Y2: PLOT X5,Y5: PLOT X3,Y3
49  COLOR =7: PLOT X1,Y1: PLOT X5,Y5: PLOT X2,Y2
50  COLOR =6: PLOT X6,Y6: PLOT X1,Y1: MUSIC "456"
51  IF X1<15 THEN 6
52  IF X2<15 THEN 6
53  IF X3<15 THEN 6
54  IF X4<15 THEN 6
55  IF X5<15 THEN 6
56  IF X6<15 THEN 6
57  PRINT "A WINNER!!": PRINT : INPUT "AGAIN Y/N",K$: IF K$="Y" THEN 5
58  PRINT "YOU TOOK "+T+" TIMES": PRINT : PRINT "THANKS FOR PLAYING"
59  END
60  PRINT "SIX FLOATING MOLECULES ARE IN ": PRINT "ONE CHAMBER OF A TWO CHAMBER
BOX"
61  PRINT "A GATE CONNECTS THE TWO TOGETHER": PRINT : PRINT "OBJECT IS TO OPEN G
ATE SO THAT"
62  PRINT "MOLECULES WILL BE ON OTHER SIDE": PRINT : PRINT "TO OPEN GATE PRESS F
IRE ON LEFT": PRINT "CL  WILL QUIT"
63  FOR R=1 TO 2500: NEXT R
64  CALL 17046,
65  COLOR =0: SHAPE =15: HLIN 12,19,6: HLIN 12,19,9: HLIN 19,28,4: HLIN 19,28,11

66  HLIN 3,12,4: HLIN 3,12,11: VLIN 4,11,3: VLIN 4,11,28
67  PLOT 12,5: PLOT 19,5: PLOT 12,10: PLOT 19,10
68  COLOR =2: SHAPE =15: HLIN 4,11,5: HLIN 4,11,6: HLIN 4,27,7
69  HLIN 4,27,8: HLIN 20,27,5: HLIN 20,27,6
70  HLIN 4,11,9: HLIN 4,11,10: HLIN 20,27,9: HLIN 20,27,10
71  COLOR =1: SHAPE =15: PLOT 15,7: PLOT 16,8: PLOT 15,8: PLOT 16,7
72  MUSIC "4444": LET X1=8:Y1=6:X2=10:Y2=7:X3=5:Y3=10:X4=5:Y4=7:X5=7:Y5=8:X6=9:Y
6=9
73  MUSIC "333": LET A=5:B=10:T=0: COLOR =3: SHAPE =15
74  PLOT X1,Y1: PLOT X2,Y2: PLOT X3,Y3: PLOT X4,Y4: PLOT X5,Y5: PLOT X6,Y6
75  RETURN
```

One thing that is helpful to Roland DeGraff of Holland MI, is to know how much memory a certain program resides in. The following program can be used to find out how much memory is used.

```
9994  DIM A$(1)
9995  FOR X=41988 TO 49151
9996  PRINT X, CHR$ ( PEEK (X))
9997 A$= KEY$ (0): IF A$="" THEN 9997
9998  NEXT X
9999  END
```

Since RAM starts out at 41988 this is where the beginning of every Basic program starts. Now, as long as you hold down the space bar or any key on the keyboard, the computer will keep printing out information in the following format: memory location----character in memory location. When you reach the last statement in the last line of your program, you simply subtract that memory location from 41988 and that will be the number of bytes that the program resides in.

On to an Auto Race program by Brett Neustater. The left joystick is used to keep the car (white) on the track (green). The game keeps track of crashes during the allotted time period. Go to it members!

```
10  REM   ...AUTO RACE...
20  DIM A$(1):X=15: FOR Q=1 TO 32: PRINT : NEXT
30  INPUT "ENTER SKILL (1=EASY 2=HARD)",V:A=13:Y=3
40  SHAPE =15: COLOR =0: FOR E=0 TO 15: HLIN 0,31,E: NEXT
50  FOR K=1 TO 200:A$= KEY$ (2)
60  IF A$="W" THEN X=X-1
70  IF A$="E" THEN X=X+1
80  COLOR =4: PLOT X,Y: IF  PEEK (512+(Y+1)*32+X)<>143 THEN 160
90  COLOR =7:U= INT ( RND (0)*3)+1: ON U GOTO 100,110,120
100 A=A+V: GOTO 120
110 A=A-V
120  IF A<1 THEN A=1
130  IF A>23 THEN A=23
140  HLIN 0,A,15: HLIN A+6,31,15: COLOR =0: HLIN A+1,A+5,15
150  PLOT X,Y: PRINT : NEXT : GOTO 170
160  CALL 17026: COLOR =7: PLOT X,Y:C=C+1: NEXT
170  PRINT "YOU CRASHED";C;"TIMES": CALL 17026
180  END
```

This next program was sent to us by Thomas Conkey of League City, TX. The program fills the screen with eight different colored stripes and can be used as a guide in getting the correct hues on your television. Thanks Thomas, for such a useful program.

```
5  CALL 17046: POKE E24578,38
10  SHAPE =15
15 X=0
20 I=0
25  COLOR =I
30  VLIN 0,15,X
35  IF X>3 THEN 45
40  GOTO 120
45  IF X>7 THEN 55
50  GOTO 120
55  IF X>11 THEN 65
60  GOTO 120
65  IF X>15 THEN 75
70  GOTO 120
75  IF X>19 THEN 85
80  GOTO 120
85  IF X>23 THEN 95
90  GOTO 120
95  IF X>27 THEN 105
100  GOTO 120
105  IF X>31 THEN 160
120 Y=Y+1
125  IF Y=4 THEN 135
130  GOTO 145
135 Y=0
140 I=I+1
145  IF I>7 THEN 160
150 X=X+1
155  GOTO 25
160  GOTO 160
```

Ah, a program for lovers, or at least one that is useful for Valentines Day. All you romantics can thank Dan Nelson of Long Beach CA for this valentine producing program.

```
10  PRINT  TAB (10);"*****"; SPC (8);"*****"
20  PRINT  TAB (8);"*********"; SPC (4);"*********"
30  PRINT  TAB (7);"***********"; SPC (2);"***********"
40  PRINT  TAB (6);"*************"; SPC (9);"*************"
60  PRINT  TAB (5);"***"; SPC (8);"HAPPY"; SPC (8);"***"
70  PRINT  TAB (5);"***"; SPC (5);"VALENTINE'S"; SPC (5);"***"
80  PRINT  TAB (5);"***"; SPC (9);"DAY"; SPC (9);"***"
100  PRINT  TAB (6);"***"; SPC (19);"***"
110  PRINT  TAB (7);"***"; SPC (17);"***"
120  PRINT  TAB (8);"***"; SPC (15);"***"
130  PRINT  TAB (9);"****"; SPC (11);"****"
140  PRINT  TAB (11);"****"; SPC (7);"****"
150  PRINT  TAB (13);"****"; SPC (3);"****"
160  PRINT  TAB (15);"*******"
170  PRINT  TAB (17);"***"
180  PRINT  TAB (18);"*";
190  GOTO 190
```

Have you ever felt as though you were being ripped off by your Gas or Electric company? From now on you can make sure that you're not; by figuring out your bill ahead of time with Peter Moranski's Electric and Gas Rates program. Nice going Peter!

```
5  GOSUB 1000
10  PRINT "ELECTRIC AND GAS RATE CALCULATOR": PRINT : PRINT
20  PRINT "IF YOU KNOW YOUR ELECTRIC AND GAS RATES TYPE 1 AND THEN RETURN, IF NO
T JUST TYPE RETURN": PRINT
30  INPUT "TYPE",A: PRINT
40  IF A=1 GOTO 240
50  GOSUB 1000
60  INPUT "PREVIOUS GAS AMOUNT $",A: PRINT
70  INPUT "PREVIOUS ELECTRIC AMOUNT $",B: PRINT
80  INPUT "GAS UNITS USED",C: PRINT
90  INPUT "KILOWATT HOURS USED",D: PRINT
100 G=A/C:E=B/D
110  GOSUB 1000
120  INPUT "PREVIOUS GAS METER READING",A: PRINT
130  INPUT "PREVIOUS ELECTRIC METER READING",B: PRINT
140  INPUT "PRESENT GAS METER READING",C: PRINT
150  INPUT "PRESENT ELECTRIC METER READING",D: PRINT
160 M=C-A:N=D-B
170 X=M*G:Y=N*E
180  GOSUB 1000
190  PRINT "YOUR GAS BILL IS $";X: PRINT
200  PRINT "YOUR ELECTRIC BILL IS $";Y: PRINT
210  PRINT "YOUR TOTAL BILL IS $";X+Y: PRINT
220  IF X+Y>250 THEN  PRINT "WOW---!"
230  END
240  INPUT "GAS---COST PER UNIT",G
250  INPUT "ELECTRIC---COST PER KILOWATT HOUR",E
260  GOTO 110
1000  CALL 17046
1010  FOR L=1 TO 32: PRINT : NEXT L
1020  POKE 40960,2: POKE 40961,0
1030  PRINT : PRINT : RETURN
```

The next two programs will probably be quite useful to many of you clubmembers. The first program was sent in by Alan Nakamoto of Richmond B.C. and it deals with square roots.

```
10  PRINT "FINDING SQUARE ROOTS"
20  PRINT
30  INPUT "NUMBER TO BE ROOTED",X
40 N=1:B=X:C=10
50  IF X>1000010000 THEN  LET C=12
60  IF X>100000100000 THEN  LET C=15
70  IF X>40000004000000 THEN  LET C=20
80  FOR A=1 TO C
90 B=(B+N)/2
100 N=X/B
110  NEXT A
120  PRINT
130  PRINT "THE SQUARE ROOT OF";X
140  PRINT "IS";B
143  PRINT B*B
145  GOTO 10
150  END
```

The second and final program for this issue was sent in by Norm Huffnayle of Wichita, KS. Norm's program converts hexidecimal numbers to decimal. (This program will be of great help to you when you're working with graphics.)

```
10  PRINT "THIS PROGRAM CONVERTS"
20  PRINT "HEXIDECIMAL NUMBERS TO DECIMAL"
30  PRINT "AND ALSO DOES THE REVERSE.": PRINT : PRINT
40  DIM N$(5),A$(3),B$(5),Z(3),D$(1),C$(1)
50  PRINT "CONVERT TO HEX(H)"
60  INPUT "OR DECIMAL(D)",D$: IF D$="D" THEN 170
70  DATA 4096,256,16,1
80  PRINT : PRINT "DECIMAL CONVERSION (99999 STOP)": PRINT
90  RESTORE : INPUT "DECIMAL NUMBER",N:Z1=N
100  IF N>99999 THEN 250
110 B$=N$:B$=" ":J=4
120  READ P:X= INT (N/P): IF X>9 THEN 140
130 Y=X+48: GOTO 150
140 Y=X+55
150 C$= CHR$ (Y):B$( LEN (B$))=C$:N=N-(P*X):J=J-1: IF J>0 THEN 120
160  PRINT "DECIMAL ";Z1;" IS ";B$;" HEX": GOTO 90
170  PRINT : PRINT 'HEX CONVERSION( "END" STOP)':J=0: PRINT
175 A$=N$
180  INPUT "HEX NUMBER",A$: IF A$="END" THEN 250
190 C$=N$:C$=A$(J):Y= ASC (C$): IF Y>64 THEN 210
200 Z(J)=Y-48: GOTO 220
210 Z(J)=Y-55
220 J=J+1: IF J<4 THEN 190
230 D=4096*Z(0)+256*Z(1)+16*Z(2)+Z(3)
240  PRINT "HEX ";A$;" IS ";D;" DECIMAL": GOTO 170
250  END
```

Well clubmembers, Norm's program wraps up this issue of the newsletter. Please don't be discouraged if your program wasn't used in this issue. So many programs are sent in, and there are so few pages to work with. Keep in mind that if we have exchanged APF software for your program, that eventually your program will be used.

Hope you enjoy this issue clubmembers; till the next issue

Editor

NEW

HARDWARE

IM-2 - IMAGINATION MACHINE	$1599.95
AL-200 - LEVEL 2 BASIC - IM-1	99.95
AL-300 - LEVEL 2 BASIC - IM-2	
VK-1 - VIDEO CONVERSION KIT	15.00

SOFTWARE

BP-50 - GENERAL LEDGER	199.95
BP-60 - COMPU-CALC (IM-2)	129.95
SY-50 - DIAGNOSTICS	15.00

.....

WHAT'S AVAILABLE IN SOFTWARE

SYSTEMS PROGRAMS

SY-20 - ASSEMBLER EDITOR (disk only)	59.95
SY-30 - BACKUP (disk only)	9.95
SY-40 - DISASSEMBLER (disk)	7.95

DISK OPERATED PROGRAMS

BP-30 - ACCOUNTS RECEIVABLE	199.95
BP-40 - MAILING LIST	29.95

COMPUTER CASSETTE PROGRAMS
(Education/Entertainment)

MA-125 - MUSIC COMPOSER/PLAYER PIANO	9.95
MA-150 - TYPING TUTOR	29.95
MA-200 - MATH TUTOR	29.95
MA-225 - PERCEPTION I	14.95
MA-325 - ARTIST and EASEL	9.95
MA-375 - THE WORD FACTORY	14.95
MA-400 - SPELLING DUEL	14.95
MA-475 - BILLBOARD/MESSAGE CENTER	19.95
MA-500 - BASIC TUTOR	49.95
MA-550 - JUMBLED UP THINGS	29.95
MA-575 - SPACE DESTROYERS	19.95

(Business/Personal Management)

MA-175 - CHECKBOOK/BUDGET MANAGER	29.95
MA-250 - SPACE, SIZE and SURFACE GUIDE	29.95
MA-275 - PERSONAL BUSINESS MACHINE	29.95
MA-300 - BUDGET MANAGER II	19.95
MA-350 - BAR CHARTS	19.95
MA-425 - ELECTRONIC FILES	29.95

*Cannot be used on IM-2

COMPUTER GAME CARTRIDGES

MG-1001 - CATENA	14.95
MG-1003 - HANGMAN/TIC TAC TOE/ DOODLE	9.95
MG-1004 - BOWLING/MICRO MATCH	19.95
MG-1005 - BRICKDOWN/SHOOTING GALLERY	9.95
MG-1006 - BASEBALL*	14.95
MG-1007 - BLACKJACK	9.95
MG-1008 - BACKGAMMON*	19.95
MG-1009 - CASINO I/ROULETTE/ KENO/SLOTS	19.95
MG-1010 - UFO/SEA MONSTERS/ BREAK IT DOWN/REBUILD/ SHOOT	19.95
MG-1011 - PINBALL/DUNGEON HUNT/ BLOCKOUT	19.95
MG-1012 - BOXING	14.95
MG-1013 - SPACE DESTROYERS	39.95

.....

WHAT'S AVAILABLE IN HARDWARE

D-100-0 MINI FLOPPY DISK DRIVE	399.95
D-100-1 2nd MINI FLOPPY DISK DRIVE	399.95
D-100-A DUAL MINI FLOPPY DISK DRIVES	799.90
R-8K - 8K RAM MEMORY CARTRIDGE (IM-1 ONLY)	79.95
R-16K - 16K RAM MEMORY CARTRIDGE (IM-2 ONLY)	99.95
WW-100 - WIRE WRAP BOARD	34.95
BB-1 - BUILDING BLOCK/SI232	159.95
BB-2 - BUILDING BLOCK/FI100	159.95
PI-100 - PARALLEL INTERFACE CARTRIDGE	99.95
SI-232 - SERIAL INTERFACE CARTRIDGE	99.95
FI-100-A - MINI FLOPPY DISK INTERFACE	99.95
TVM-10 - VIDEO MONITOR	149.95

.....

WHAT'S AVAILABLE IN DOCUMENTATION

TRM - TECHNICAL REFERENCE MANUAL	N/C
LBL-1 - LISTINGS, I/O ROUTINES, DOS	14.95
LBL-2 - LISTINGS, BASIC INTERPRETER	19.95
SM-1 - IM-1 SERVICE MANUAL	14.95

STAR TREX***

THE ULTIMATE STRATEGIC SPACE BATTLE!

SEEK OUT AND DESTROY THE FLEET OF KLINGON WARSHIPS WHICH THREATEN THE GALAXY IN A RACE AGAINST "TIME", AS YOU COMMAND YOUR STARSHIP'S:



SHORT RANGE SCAN GIVES SEMI-GRAPHICS DISPLAY OF YOUR STARSHIP'S LOCATION AND ALL STARS, KLINGONS AND STARBASES WITHIN YOUR PRESENT QUADRANT. STAR TREX IS NOT AN "ARCADE" GAME, RATHER AN EXCITING GAME OF SKILL AND STRATEGY. REMEMBER TO KEEP YOUR SHIELDS UP, AS THE KLINGONS' PHASERS CAN DAMAGE ANY PART OF YOUR STARSHIP WITH SURPRISING SPECIAL EFFECTS!

STAR TREX INTRO PROGRAM AND GAME PROGRAM .... 8K $12.95



NO MORE LOST OR BENT CARDS! LET YOUR IMAGINATION MACHINE SHUFFLE AND DEAL OUT THE 52 CARD DECK AS YOU AND YOUR OPPONENT USE THE JOYSTICK/CONTROLLERS TO TRY AND SELECT PAIRS. LEFT AND RIGHT SCORES ARE ALSO DISPLAYED ON THE SCREEN IN THIS HI-RES GAME.



YOUR COMPUTER KEEPS SCORE AS 2 PLAYERS USE THE JOYSTICK/CONTROLLERS TO EVADE, PURSUE AND SHOOT THE HEADS IN THIS FAST-ACTION, HI-RES GAME. WATCH OUT, THERE'S ONE BEHIND YOU!

CONCENTRATION AND HEAD-HUNTER (BOTH) ..... 8K ..... $ 9.95



ENTER THE MINOTAUR'S CASTLE, DESTROY IT AND CLAIM IT'S TREASURES IN THIS ADVENTURE GAME. (NO GRAPHICS).

MINOTAUR ..... 8K ..... FREE WITH PURCHASE OF BOTH TAPES!

IMAGINATION MACHINE PROGRAMS ABOVE MAY BE PURCHASED FROM:

G. R. JONES
419 S. 105 E. PL.
TULSA, OKLAHOMA 74128

ON CASSETTE W/BOX

ALL 4 FOR $22.9

MONEY ORDERS (OR CHECKS) ONLY, PLEASE.

PROGRAMS FOR FUN AND LEARNING FOR APF COMPUTERS

Program Numbers

TIC TAC KNOW

#101 8 K VERSION - Two players take turns trying to claim the boxes on a tic-tac-toe game board. The boxes are won by correctly answering questions. Input is via the game controllers. There are 9 topics with 4 questions on each topic.
$8.00

#102 16 K VERSION - The game board is the same as the 8K version except that a blue and redsquare is used to mark the boxes. Also, there are a total of 81 questions; 9 topics with 9 questions each. This version has the ability to recognize when someone has won. It then ends the game and states who won.
$12.00

TIME MACHINE

#103 16 K - This is a fun way to learn history. The game is for 2,3 or 4 players. At the onset of the game you enter the number of rounds to be played and the level of difficulty for each player. You also decide whether or not to have the correct answer shown. Computer keeps score and keeps track of the turns. Players use game controllers.
$12.00

MATH TIME

#104 8 K - Ever have trouble reading numbers on the screen ??? Well, all of the numbers in this program are in large, colorful graphics. With this program you can practice addition, subtraction, multiplication ,division,times tables, and a mixture of operations. The computer keeps track of your score and amount of time needed. Try to increase your speed and accuracy.
$8.00

STAR BATTLE

#105 8 K - One player at the keyboard. Sorry, no graphics. You have to find and destroy the enemy space ship. You have a limited amount of fuel. Be careful, the enemy will occassionally shoot at you and he may also change his position.
$5.00

MONEY

#106 8 K - This program is designed to help children to learn how to add up money and make change. The child is told the amount of money he has and then is given the chance to spend it and compute his change.
$5.00

SYMBOLS

#107 8 K - This program is for anyone interested in science or planning a career in science. The program will help a person to learn the symbols and names of 25 common chemical elements. Symbols are displayed as large, colorful low-resolution characters.
$8.00

CODE PRACTICE 8 K - Learn and practice morse code. Allows you to practice random letters or groups of letters. Good for those interested in radio electronics.
#108 $8.00

_

TO ORDER ONE OR MORE PROGRAMS SEND A CHECK OR MONEY ORDER FOR THE APPROPRIATE AMOUNT ALONG WITH THE PROGRAM NUMBER(S) TO

Richard Carman, 7 Roda Dr. , Mastic,N.Y. 11950

MAKE CHECKS AND MONEY ORDERS PAYABLE TO Richard Carman.

PROFESSIONALLY CREATED PROGRAMS FOR THE APF IMAGINATION MACHINE

EXECUTIVE SOFTWARE AVAILABLE:

CONTROLLED LIST. Scroll back and forth through a BASIC program using the joystick. Make changes and resume scrolling. An ideal tool for debugging or modifying a program.

RENUMBER. Renumber an APF BASIC program, leaving 5 line intervals between statements to allow for modifications. NO IN-LINE CODE ALLOWED.

GRAPHICS HELPER. HI-RES programers can now create shapes using the joystick. This program creates the codes necessary to build the design, displays it and allows modifications to any created shapes. The menu allows the programer to list the codes generated for each shape in HEX and DEC and to display any designs on the screen.
BUT THE BEST PART is that it provides for the shapes created to be used in another BASIC program as a REMARK statement. A CALL routine is provided which relocates the shape table. The only coding needed to put HI-RES shapes on the screen is POKE Address, Shape. A MUST FOR THE SERIOUS HI-RES PROGRAMMER!

DISK-TO-TAPE BACKUP/RESTORE. Backup disks to tape and restore tape to disk. Eliminate the need for 2 disk units or a lot of manual switching of disks. Once started NO OPERATOR INTERVENTION is required. Start the program and take a break! LET THE COMPUTER DO THE WORK.

GAMES FOR THE APF:

APF-MAN. FINALLY an APF version of this most popular arcade game is available! The little muncher scurries around the screen trying to gobble up as many points as he can while being pursued by some very intelligent robots. Each new screen becomes more challenging as the robots pick up speed. A VERY GOOD HI-RES SIMULATION OF THE ARCADE GAME, complete with bonus points and energizers. The most REALISTIC game for the APF since SPACE INVADERS!

TREASURE HUNT. Accumulate as much treasure as possible as quickly as you can by maneuvering around the board with the joystick. Two players may pit their skills against each other or take on the computer in one of five skill levels. Each game is different so strategy is a must.

JIGSAW PUZZLES. Program builds jigsaw puzzles to be solved in as few moves as possible. Every puzzle is different. It's not as simple as you might think.

ART THIEF. A valuable painting has been stolen and you must determine who stole it and when. Question the suspects and deduce the solution. Be careful though because some people are forgetful and the thief, of course, might lie. LO-RES graphics game with 5 skill levels.

GRAPHICS HELPER	$ 15	APF-MAN	$ 20
RENUMBER	8	TREASURE HUNT	8
CONTROLLED LIST	8	JIGSAW PUZZLES	8
TAPE/DISK BACKUP	8	ART THIEF	8
SOFTWARE PACK	35	GAME PACK	40
{INCLUDES ALL OF ABOVE}		{INCLUDES ALL OF ABOVE}	

SEND CHECK/MONEY ORDER TO: CARL A. ECHOLS
112 CREEKSIDE LANE
NOBLESVILLE, IN 46060

NAME----------

STREET----------

CITY---------- STATE---------- ZIP-------